Chow forms

Chris Peterson and Hirotachi Abo

Note. This script is also available at:

http://www.math.colostate.edu/~abo/Research/smi/smi-algebraic-geometry.html

1. Chow form of a line in $\mathbb{P}^3$

Let k be an algebraically closed field, let $\mathbb{P}^3$ be the three-dimensional projective space over k. We denote by S the homogeneous coordinate ring $k[x_0, x_1, x_2, x_3]$ of $\mathbb{P}^3$. Fix a line L in $\mathbb{P}^3$. The general line in $\mathbb{P}^3$ does not intersect L. So the lines in $\mathbb{P}^3$ hitting L form a proper subset $C(L)$ (actually a subvariety) of the grassmaniann of lines in $\mathbb{P}^3$. A question is: "How can we describe this subset?" Assume that the ideal $I(L)$ of L is generated by the following two linear forms: $\sum_{i=0}^3 a_{0i}x_i$ and $\sum_{i=0}^3 a_{1i}x_i$. Let L' be a line in $\mathbb{P}^3$ defined by linear forms $\sum_{i=0}^3 b_{0i}x_i$ and $\sum_{i=0} b_{1i}x_i$. Then L and L' intersect if and only if the determinant of the matrix

$$\Lambda = \begin{pmatrix} a_{00} & a_{01} & a_{02} & a_{03} \\ a_{10} & a_{11} & a_{12} & a_{13} \\ b_{00} & b_{01} & b_{02} & b_{03} \\ b_{10} & b_{11} & b_{12} & b_{13} \end{pmatrix}$$

is zero. For $0 \le i < j \le 3$, let

$$\lambda_{ij} = a_{ij}a_{i+1j+1} - a_{i+1j}a_{ij+1} \text{ and } \Lambda_{ij} = b_{ij}b_{i+1j+1} - b_{i+1j}b_{ij+1}.$$

From Problem 3 in Problem Set 20, it follows that $\det(\Lambda) = 0$ if and only if

$$F = \lambda_{01}\Lambda_{23} - \lambda_{02}\Lambda_{13} + \lambda_{03}\Lambda_{12} + \lambda_{12}\Lambda_{03} - \lambda_{13}\Lambda_{02} + \lambda_{23}\Lambda_{01} = 0.$$

Recall that the Plücker embedding from $\mathbb{G}(1,3)$ to $\mathbb{P}^5$ is defined by

$$L' \mapsto [\Lambda_{01} : \Lambda_{02} : \Lambda_{12} : \Lambda_{03} : \Lambda_{13} : \Lambda_{23}] = [X_0 : \cdots : X_5].$$

Consider the ring $k[b_{00}, \ldots, b_{13}, X_0, \ldots, X_5]$ and the ideal

$$I = (X_0 - \Lambda_{01}, X_1 - \Lambda_{02}, X_2 - \Lambda_{12}, X_3 - \Lambda_{03}, X_4 - \Lambda_{13}, X_5 - \Lambda_{23}, F).$$

Let $J = I \cap k[X_0, \ldots, X_5]$. Then $J = (Q, F')$, where

$$Q = X_0X_5 - X_1X_4 + X_2X_3 \tag{1}$$

and

$$F' = \lambda_{01}X_5 - \lambda_{02}X_4 + \lambda_{12}X_3 + \lambda_{03}X_2 - \lambda_{13}X_1 + \lambda_{23}X_0. \tag{2}$$

Recall that Q is the defining equation of $\mathbb{G}(1,3)$. So $C(L)$ can be regarded as a hypersurface in $\mathbb{G}(1,3)$. This hypersurface is called the *Chow variety* of L, and the linear form F' is called the *Chow form* of L. For a line L in $\mathbb{P}^3$ chosen at random, we compute the Chow form with `Macaulay2`:

```
i1 : KK=QQ;

i2 : ringP3=KK[x_0..x_3];

i3 : L=ideal random(ringP3^{0},ringP3^{2:-1})

            5             8      2      2
o3 = ideal (-*x  + x , - -*x  - -*x  - -*x )
            2  1    2    9  0   5  1   3  2

o3 : Ideal of ringP3

i4 : coeff=transpose diff(transpose (vars ringP3),gens L)

o4 = {-1} | 0    5/2  1    0 |
     {-1} | -8/9 -2/5 -2/3 0 |

                   2             4
o4 : Matrix ringP3  <--- ringP3
```

Using (2), we can compute the chow form of L:

$$F' = -\frac{19}{15}x_3 - \frac{8}{9}x_4 + \frac{20}{9}x_5.$$

Let's check this!

```
i5 : ringP7=KK[b_(0,0)..b_(1,3)];

i6 : mat=matrix{{b_(0,0)..b_(0,3)},{b_(1,0)..b_(1,3)}}

o6 = | b_(0,0) b_(0,1) b_(0,2) b_(0,3) |
     | b_(1,0) b_(1,1) b_(1,2) b_(1,3) |
```

```
                  2            4
o6 : Matrix ringP7  <--- ringP7

i7 : pluecker=minors(2,mat)

o7 = ideal (- b   b    + b   b   , - b   b    + b   b   , - b   b    + b   b   ,
               0,1 1,0    0,0 1,1     0,2 1,0    0,0 1,2     0,2 1,1    0,1 1,2

o7 : Ideal of ringP7

i8 : gamma=substitute(coeff,ringP7)||mat

o8 = {-1} | 0       5/2     1       0       |
     {-1} | -8/9    -2/5    -2/3    0       |
     {0}  | b_(0,0) b_(0,1) b_(0,2) b_(0,3) |
     {0}  | b_(1,0) b_(1,1) b_(1,2) b_(1,3) |

                  4            4
o8 : Matrix ringP7  <--- ringP7

i9 : F=det gamma

     19            8             20             19            8            20
o9 = --*b   b    + -*b   b    - --*b   b    - --*b   b    - -*b   b    + --*b
     15  0,3 1,0   9  0,3 1,1    9  0,3 1,2   15  0,0 1,3   9  0,1 1,3    9  0,2

o9 : ringP7

i10 : ringP5=KK[X_0..X_5];

i11 : ringP7xP5=KK[b_(0,0)..b_(1,3),X_0..X_5,Degrees=>{8:1,6:2},
        MonomialOrder=>Eliminate 8];

i12 : grass=substitute(vars ringP5,ringP7xP5)-
         substitute(gens pluecker,ringP7xP5);

                      1               6
o12 : Matrix ringP7xP5  <--- ringP7xP5

i13 : hyper=substitute(F,ringP7xP5)
```

```
      19            8            20            19            8            20
o13 = --*b    b    + -*b    b    - --*b    b    - --*b    b    - -*b    b    + --*b
      15  0,3 1,0    9  0,3 1,1    9  0,3 1,2    15  0,0 1,3    9  0,1 1,3    9  0,

o13 : ringP7xP5

i14 : gr=ideal grass+ideal hyper;

o14 : Ideal of ringP7xP5

i15 : chowVariety=ideal substitute(selectInSubring(1,gens gb gr),ringP5)

                 40      100            40              100
o15 = ideal (X  + --*X  - ---*X , X X  + --*X X  - X X  - ---*X X )
              3   57  4    57  5   1 4   57  2 4    0 5    57  2 5

o15 : Ideal of ringP5
```

The qudratic polynomial Q in `chowVariety` looks a little different from (1). But this polynomial was just reduced. Indeed,

$$
\begin{aligned}
Q &= X_1X_4 + \frac{40}{57}X_2X_4 - X_0X_5 - \frac{100}{57}X_2X_5 \\
&= X_1X_4 - X_0X_5 - X_2\left(-\frac{40}{57}X_4 + \frac{100}{57}X_5\right) \\
&= X_1X_4 - X_0X_5 - X_2X_3.
\end{aligned}
$$

So Q differs from (1) by the sign.

```
i16 : chowForm=(gens chowVariety)_(0,0)

           40      100
o16 = X  + --*X  - ---*X
       3   57  4    57  5

o16 : ringP5
```

The linear form we have obtained is equal to $-\frac{15}{19}F'$.

2. Number of 4-secant lines to four skew lines in $\mathbb{P}^3$

We start with the following question:

Question 1. Let L_1, L_2, L_3 and L_4 be four skew lines in $\mathbb{P}^3$. Is there a line which intersects all of them? If such a line exists, are there finitely many such lines or infinite many?

Suppose that there exists such a line L. Then L can be regarded as a point in $\mathbb{G}(1,3)$. Since L hits L_1, L_2, L_3 and L_4, the corresponding point in $\mathbb{G}(1,3)$ is contained in $C(L_1) \cap C(L_2) \cap C(L_3) \cap C(L_4)$. Let F_i denote the Chow form of L_i, $i = 1, 2, 3, 4$. Then the intersection of the Chow varieties is defined by the ideal $I = (Q, F_1, F_2, F_3, F_4)$, where Q is the defining equation of $\mathbb{G}(1,3)$ in $\mathbb{P}^5$. Since the ideal is generated by five polynomials, the corresponding variety $V(I)$ cannot be empty. From the generality of the choice of the four skew lines, we can expect that $\{Q, F_1, F_2, F_3, F_4\}$ is a minimal generating set for I. In this case, $\dim(V(I)) = 0$, that is, $V(I)$ is a finite set of points. The next question is therefore:

Question 2. How many points are there in $V(I)$?

Recall that the "degree" of a given hypersurface in $\mathbb{P}^n$ is defined to be the intersection number of the hypersurface itself and the general line in $\mathbb{P}^n$. On the other hand, the degree of an r-dimensional projective variety in $\mathbb{P}^n$ is defined to be $r!$ times the leading coefficient of its Hilbert polynomial (see Chapter I-7 in *Algebraic Geometry* by R. Hartshorne). The Hilbert polynomial P_V of a hypersurface V in $\mathbb{P}^n$ can be easily computed. Let R be the homogeneous coordinate ring of $\mathbb{P}^n$. If the polynomial defining V has degree d, then P_V is obtained from the exact sequence:

$$0 \to R(-d) \to R \to \Gamma(V) \to 0.$$

Indeed, we obtain

$$P_V(t) = \binom{n+t}{n} - \binom{n+t-d}{n} = \frac{d}{(n-1)!}t^{n-1} + \cdots .$$

So $\deg(V) = d$. This implies that $\deg(\mathbb{G}(1,3)) = 2$, because $\deg(Q) = 2$. The Chow forms F_1, F_2, F_3 and F_4 define a line in $\mathbb{P}^5$, and this line meets $V(Q)$ exactly in two points, because otherwise the line would lie on $V(Q)$ and there are infinitely many lines which intersect all four lines. But this contradicts our assumption. Therefore the number of points in $V(I)$ is expected to be 2.

Exercise (Problem 1 in Problem Set 21). Given four skew lines in $\mathbb{P}^3$, show that the number of lines which intersect all of them is equal to 2.

Hint. Use either Formula (2) or the `Macaulay2` script to get the Chow forms of the four lines.